

3.6: Duhamel's Theorem

3.6.1 Introduction

The separation of variables technique discussed in Section 3.5 is not capable of solving problems with time-dependent terms (e.g., an ambient temperature or applied heat flux that varies with time). However, problems with time dependent terms are common. Duhamel's theorem provides one method of extending an analytical solution that is derived (for example, using separation of variables) assuming a time invariant term in order to consider the temperature response to an arbitrary time variation of that term.

3.6.2 Duhamel's Theorem

It is somewhat easier to state Duhamel's theorem than it is understand it; Duhamel's theorem says that:

If $T_f(x,t)$ is the response of a linear system with a zero initial condition to a single, constant non-homogeneous term with magnitude of unity (referred to as the fundamental solution), then the response of the same system to a single, time-varying non-homogeneous term with magnitude $B(t)$ can be obtained from the fundamental solution according to:

$$T(x,t) = \int_{\tau=0}^t T_f(x,t-\tau) \frac{dB(\tau)}{d\tau} d\tau + B_{t=0} T_f(x,t) \quad (3-401)$$

where $B_{t=0}$ is the value of B at time zero and B must be continuous in time.

In order to apply Duhamel's theorem, it is necessary to have a problem with a zero initial condition and a single non-homogeneous term that varies in time. The problem must be divided into sub-problems so that this is the case. Once this has been accomplished, it is necessary to obtain the fundamental solution, T_f , to the sub-problem with the time-varying term replaced by a constant value, 1. Finally, Duhamel's theorem can be applied to the fundamental solution, according to Eq. (3-401). The integration in Eq. (3-401) can be accomplished either analytically or numerically.

There are a number of steps that must be carried out in order to apply Duhamel's theorem. The process is illustrated in this section in the context of the problem that was considered in Section 3.5.5, and is re-stated here with a time-dependent boundary condition. A plane wall (see Figure 3-31) is initially at temperature $T_{ini} = 20^\circ\text{C}$ when at time $t = 0$ the surface of the wall (at $x = L$) is subjected to a heat flux that varies in time according to a Gaussian function:

$$\dot{q}_s''(t) = q_{max}'' \exp\left[-\frac{(t-t_p)^2}{2t_d^2}\right] \quad (3-402)$$

E9: Section 3.6 Duhamel's Theorem

where $q''_{max} = 5000 \text{ W/m}^2$ is the maximum heat flux, $t_p = 200 \text{ s}$ is the time associated with the peak heat flux, and $t_d = 50 \text{ s}$ is the duration of the peak. Equation (3-402) is shown in Figure 3-33. The left side of the wall (at $x = 0$) is maintained at $T_s = T_{ini} = 20^\circ\text{C}$ for all time. The wall is $L = 0.1 \text{ m}$ thick and made of material with $k = 10 \text{ W/m-K}$ and $\alpha = 1 \times 10^{-4} \text{ m}^2/\text{s}$.




Figure 3-33: Heat flux at the right side of the wall as a function of time.

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA K J
$TABSTOPS 0.2 0.4 0.6 0.8 3.5 in
```

"Inputs"

T_ini=converttemp(C,K,20 [C])

"initial temperature"

T_s=converttemp(C,K,20 [C])

"surface temperature"

q``_dot_max=5000 [W/m^2]

"maximum heat flux"

t_d=50 [s]

"duration of peak"

t_p=200 [s]

"time of peak"

k=10 [W/m-K]

"conductivity"

alpha=1e-4 [m^2/s]

"thermal diffusivity"

L=0.1 [m]

"thickness of the wall"

The governing partial differential equation for the problem is:

$$\frac{\partial^2 T}{\partial x^2} = \frac{1}{\alpha} \frac{\partial T}{\partial t} \quad (3-403)$$

The boundary conditions for the problem are:

$$T_{x=0} = T_{ini} \quad (3-404)$$

$$k \left. \frac{\partial T}{\partial x} \right|_{x=L} = q''_{max} \exp \left[-\frac{(t-t_p)^2}{2t_d^2} \right] \quad (3-405)$$

$$T_{t=0} = T_{ini} \quad (3-406)$$

Isolate the Time-Dependent Term

In order to apply Duhamel's theorem, it is necessary that the problem have a zero initial condition and that the time dependent term be the only non-homogeneous one. In most problems this is not the case. Therefore, it will be necessary to assume that the solution is the sum of the solutions to several sub-problems; one of these is defined so that it has these characteristics. This process is achieved by following the steps outlined in Section 2.3.2 or Section 3.5.5. The sub-problems can then be solved using separation of variables in addition to the one that will be tackled with Duhamel's theorem.

The problem discussed above can be transformed by inspection in order to isolate the time-dependent boundary condition by defining the temperature difference relative to T_{ini} :

$$\theta = T - T_{ini} \quad (3-407)$$

The transformed partial differential equation is:

$$\frac{\partial^2 \theta}{\partial x^2} = \frac{1}{\alpha} \frac{\partial \theta}{\partial t} \quad (3-408)$$

The boundary conditions for the transformed problem are:

$$\theta_{x=0} = 0 \quad (3-409)$$

$$k \left. \frac{\partial \theta}{\partial x} \right|_{x=L} = q_{max}'' \exp \left[-\frac{(t-t_p)^2}{2t_d^2} \right] \quad (3-410)$$

$$\theta_{t=0} = 0 \quad (3-411)$$

Equations (3-408) through (3-411) satisfy the criteria for applying Duhamel's theorem.

Obtain the Fundamental Solution

The fundamental solution, θ_f , is the solution to the isolated sub-problem with the time-dependent term replaced by a constant, unit value. For this problem, the time-dependent heat flux in Eq. (3-410) is replaced by unity. The mathematical specification of the fundamental problem is therefore:

$$\frac{\partial^2 \theta_f}{\partial x^2} = \frac{1}{\alpha} \frac{\partial \theta_f}{\partial t} \quad (3-412)$$

$$\theta_{f,x=0} = 0 \quad (3-413)$$

E9: Section 3.6 Duhamel's Theorem

$$k \frac{\partial \theta_f}{\partial x} \Big|_{x=L} = 1 \quad (3-414)$$

$$\theta_{f,t=0} = 0 \quad (3-415)$$

The fundamental problem will, by definition, include at least one non-homogeneous term and therefore it is likely that the techniques discussed in Section 3.5.5 must be applied. The solution to the problem posed by Eqs. (3-412) through (3-415) was obtained in Section 3.5.5 (with $T_{ini} = T_s$ replaced by 0 and \dot{q}_s'' replaced by 1). The fundamental solution is obtained from Eq. (3-400):

$$\theta_f = \sum_{i=1}^{\infty} C_i \sin(\lambda_i x) \exp(-\lambda_i^2 \alpha t) + \frac{x}{k} \quad (3-416)$$

where the eigenvalues are provided by Eq. (3-396):

$$\lambda_i = \frac{(2i-1)\pi}{2L} \quad (3-417)$$

and the constants are provided by Eq. (3-399):

$$C_i = -\frac{8L}{k\pi^2} \frac{(-1)^{1+i}}{(4i^2 - 4i + 1)} \quad (3-418)$$

The fundamental solution is programmed in EES:

```
"fundamental solution"
x_hat=0.75 [-]
x=x_hat*L
time=20 [s]
N_term=10 [-]
duplicate i=1,N_term
  lambda[i]=(2*i-1)*pi/(2*L)
  C[i]=-8*L*(-1)^(1+i)/(k*pi^2*(4*i^2-4*i+1))
  theta_f[i]=C[i]*sin(lambda[i]*x)*exp(-lambda[i]^2*alpha*time)
end
theta_f=SUM(theta_f[1..N_term])+x/k
```

"dimensionless position"
"position"
"time"
"number of terms"
"eigenvalue"
"constant"
"solution for i'th eigenvalue"
"fundamental solution"

The fundamental solution should be checked to ensure that it agrees with your intuition. The fundamental solution represents the temperature response per unit of disturbance (in this case, per unit of heat flux) and therefore has units $\text{K}\cdot\text{m}^2/\text{W}$. The fundamental solution should begin from a zero initial condition with a zero temperature at the left-hand side and the material should then respond qualitatively as it did in Section 3.5.5. Figure 3-34 illustrates the fundamental solution as a function of position for various values of time.

E9: Section 3.6 Duhamel's Theorem




Figure 3-34: Fundamental solution as a function of position for various values of time.

Apply Duhamel's Theorem

The fundamental solution can be substituted into Duhamel's theorem, Eq. (3-401), with $\dot{q}_s''(t)$ in place of $B(t)$:

$$\theta(x,t) = \int_{\tau=0}^t \theta_f(x,t-\tau) \frac{d\dot{q}_s''(\tau)}{dt} d\tau + \dot{q}_{s,t=0}'' \theta_f(x,t) \tag{3-419}$$

Figure 3-33 shows that the value of the heat flux at $t = 0$ is very near zero and so the second term in Eq. (3-419) can be neglected:

$$\theta(x,t) = \int_{\tau=0}^t \theta_f(x,t-\tau) \frac{d\dot{q}_s''(\tau)}{dt} d\tau \tag{3-420}$$

Equation (3-420) can be integrated either analytically or numerically. Here we will apply Duhamel's theorem using numerical integration with EES' Integral function. Before carrying out the integral, it is necessary to ensure that the integrand of Eq. (3-420) can be evaluated, given an arbitrary value of the integration variable, τ . The additional EES code is shown below in bold:

```
"fundamental solution"
x_hat=0.75 [-]                                "dimensionless position"
x=x_hat*L                                     "position"
time=20 [s]                                    "time"
tau = 2 [s]                                  "integration variable"
N_term=10 [-]                                  "number of terms"
duplicate i=1,N_term
  lambda[i]=(2*i-1)*pi/(2*L)                   "eigenvalue"
  C[i]=-8*L*(-1)^(1+i)/(k*pi^2*(4*i^2-4*i+1)) "constant"
  theta_f[i]=C[i]*sin(lambda[i]*x)*exp(-lambda[i]^2*alpha*(time-tau))"solution for i'th eigenvalue"
end
theta_f=SUM(theta_f[1..N_term])+x/k           "fundamental solution at t - tau"
```

The derivative of the heat flux at the surface, Eq. (3-402), is evaluated symbolically using Maple:

```
> restart;
```

E9: Section 3.6 Duhamel's Theorem

> $qf:=qfmax*exp(-(t-t_p)^2/(2*t_d^2));$

$$qf := qfmax e^{\left(-\frac{(t-t_p)^2}{2t_d^2}\right)}$$

> $dqfdt:=diff(qf,t);$

$$dqfdt := -\frac{qfmax (t-t_p) e^{\left(-\frac{(t-t_p)^2}{2t_d^2}\right)}}{t_d^2}$$

$$\frac{dq_s''(t)}{dt} = -\frac{q_{max}'' (t-t_p)}{t_d^2} \exp\left[-\frac{(t-t_p)^2}{2t_d^2}\right] \quad (3-421)$$

and evaluated in EES at time τ .

$dq_dot_dt=-q_dot_max*(tau-t_p)*exp(-(tau-t_p)^2/(2*t_d^2))/t_d^2$ "derivative of heat flux"

Both terms in the integral of Eq. (3-420) can be calculated, given a value of τ . Therefore, the specification of τ is commented out and the integral in Eq. (3-420) is carried out using the Integral function in EES:

$theta=INTEGRAL(theta_f*dq_dot_dt,tau,0,time)$ "Duhamel's theorem"

The solution for temperature is obtained using Eq. (3-407) and converted to Celsius:

$T=theta+T_ini$ "temperature"
 $T_C=converttemp(K,C,T)$ "in C"

A parametric table is created in which time is varied from 0 to 500 s and used to prepare Figure 3-35.




Figure 3-35: Temperature as a function of time for various values of position.